... but there are no ERROR messages in the LOGs

Shafi Chowdhury, Shafi Consultancy Limited, London, U.K.

ABSTRACT
SAS® is such a versatile tool that it can allow users to do pretty much what they want. The only problem with this flexibility is that users can make mistakes and not realize that there is an error in their code. Many people just search for the word ERROR before assuming that all is well. How often do we see “uninitialized”, “repeats of BY” or “W.D format was too small” in the log and ignore them? This paper will look at some of the mistakes people make in their everyday programming which do not result in an ERROR message in the LOG, but may result in incorrect numbers. This paper will also show how we can identify these hidden errors and resolve them.

INTRODUCTION
SAS® is used across the world in many industries for many different tasks. It is used such widely because it is so versatile, it can be easily adapted for the different situations where it is used. The only problem with this versatility is that the user then has to take care that what they do makes sense for their purpose. Something can be considered acceptable in one situation and not in another; the user has to make the decision. This is all well and good provided the user is aware of what they need to look out for, and this is often the weakest point.

Often users just search their LOG files for the word ERROR, and if they are a little more vigilant then they will also search for the word WARNING. However, these are not the only indicators to suggest that there may be errors in the result. These other indicators or “silent errors” can produce incorrect results or produce correct results but highlight data issues without any ERROR or WARNING messages appearing in the LOG. Most of the programs written in the pharmaceutical industry are data driven to a certain extent. Investigators love to challenge us and they find ever more innovative ways to produce incorrect results. This means the programmers have to be on their guard to ensure bad data is identified and dealt with in a pre-specified manner.

Defensive programming is described in Good Programming Practice guides across the industry. This is where known “unwanted” messages are eliminated from the LOG using conditional programming codes and then sending user specified messages to the LOG to highlight bad data. However, there are too many different possibilities as to what might be considered an error, so some of the most common ones from the pharmaceutical industry are presented here.
UNINITIALIZED
This is a common problem. It can be the result of a typo or simply using a variable which has not been defined yet. It means a variable which has not yet been initialised in the data step is being used in a calculation. As SAS does not know the variable, a missing value is assigned to it during the calculation. This assumption of a missing value can lead to further consequences, some of which may not be intended. As a result it is always good practice to initialise variables prior to them being used in calculations. This way, at least the user will know that if a missing value was used then it was missing on purpose and not as a result of an oversight.

Program:
```sas
DATA un_ini;
    a=12;
    b=10;
    c=a*d;
RUN;
```

Log:
```text
NOTE: Variable D is uninitialized.
NOTE: Missing values were generated as a result of performing an operation on missing values.
    Each place is given by: (Number of times) at (Line):(Column).
    1 at 5:6
NOTE: The data set WORK.UN_INI has 1 observations and 4 variables.
```

Output:
```text
Obs   A   B   C   D
1     12  10  .  .
```

In this example it should have been c=a*b, but a typo resulted in c=a*d. The final consequence of this is that where as it should have resulted in c=120, the result now is c=. (missing) and there is an additional variable D which also has a missing value.
REPEATS OF BY
This perhaps is not as common, but it still appears in many LOG files and the users are not aware that its effect could be a problem. When we are merging two datasets often one of them has more than one record and the other has one record per by variable combination. The result is that the dataset with one record is merged to all the records for that by variable combination in the second dataset to create a new dataset. However, if one patient has two records instead of one for some reason, then the result is very different for that patient. Unless the message in the log is recognised, this problem can go unnoticed and lead to errors in things like calculating change from baseline.

Program:

* Create dummy data, patient 1 has two visit 1 records *
DATA wt_by_vis;
  DO ptno=1 TO 3;
    DO visit=1 TO 3;
      weight=60+MOD(INT((RANUNI(0)*1000)),40);
      OUTPUT;
      IF ptno=1 AND visit=1 THEN DO;
        weight=60+MOD(INT((RANUNI(0)*1000)),40);
        OUTPUT;
      END;
    END;
  END;
RUN;
TITLE3 'dataset 1 - Weight by visit';
PROC PRINT DATA=wt_by_vis;
RUN;
TITLE3;

* Keep all visit 1 records as baseline *
DATA baseline;
  SET wt_by_vis;
  WHERE visit=1;
RUN;

* Merge baseline to the rest of the data and calculate change from baseline *
DATA change;
  MERGE wt_by_vis baseline(KEEP=ptno weight RENAME=(weight=baseline));
  BY ptno;
  change=weight-baseline;
RUN;
TITLE3 'dataset 2 - CHANGE from baseline';
PROC PRINT DATA=change;
RUN;
TITLE3;

Log:
NOTE: The data set WORK.WT_BY_VIS has 10 observations and 3 variables.
NOTE: There were 10 observations read from the data set WORK.WT_BY_VIS.
NOTE: The data set WORK.BASELINE has 4 observations and 3 variables.
NOTE: MERGE statement has more than one data set with repeats of BY values.
NOTE: There were 10 observations read from the data set WORK.WT_BY_VIS.
NOTE: There were 4 observations read from the data set WORK.BASELINE.
NOTE: The data set WORK.CHANGE has 10 observations and 5 variables.

NOTE: There were 10 observations read from the data set WORK.CHANGE.

Output:

dataset 1 - Weight by visit

<table>
<thead>
<tr>
<th>Obs</th>
<th>PTNO</th>
<th>VISIT</th>
<th>WEIGHT</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1</td>
<td>1</td>
<td>77</td>
</tr>
<tr>
<td>2</td>
<td>1</td>
<td>1</td>
<td>73</td>
</tr>
<tr>
<td>3</td>
<td>1</td>
<td>2</td>
<td>66</td>
</tr>
<tr>
<td>4</td>
<td>1</td>
<td>3</td>
<td>88</td>
</tr>
<tr>
<td>5</td>
<td>2</td>
<td>1</td>
<td>90</td>
</tr>
<tr>
<td>6</td>
<td>2</td>
<td>2</td>
<td>81</td>
</tr>
<tr>
<td>7</td>
<td>2</td>
<td>3</td>
<td>68</td>
</tr>
<tr>
<td>8</td>
<td>3</td>
<td>1</td>
<td>97</td>
</tr>
<tr>
<td>9</td>
<td>3</td>
<td>2</td>
<td>60</td>
</tr>
<tr>
<td>10</td>
<td>3</td>
<td>3</td>
<td>96</td>
</tr>
</tbody>
</table>

dataset 2 - CHANGE from baseline

<table>
<thead>
<tr>
<th>Obs</th>
<th>PTNO</th>
<th>VISIT</th>
<th>WEIGHT</th>
<th>BASELINE</th>
<th>CHANGE</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1</td>
<td>1</td>
<td>77</td>
<td>77</td>
<td>0</td>
</tr>
<tr>
<td>2</td>
<td>1</td>
<td>1</td>
<td>73</td>
<td>73</td>
<td>0</td>
</tr>
<tr>
<td>3</td>
<td>1</td>
<td>2</td>
<td>66</td>
<td>73</td>
<td>-7</td>
</tr>
<tr>
<td>4</td>
<td>1</td>
<td>3</td>
<td>88</td>
<td>73</td>
<td>15</td>
</tr>
<tr>
<td>5</td>
<td>2</td>
<td>1</td>
<td>90</td>
<td>90</td>
<td>0</td>
</tr>
<tr>
<td>6</td>
<td>2</td>
<td>2</td>
<td>81</td>
<td>90</td>
<td>-9</td>
</tr>
<tr>
<td>7</td>
<td>2</td>
<td>3</td>
<td>68</td>
<td>90</td>
<td>-22</td>
</tr>
<tr>
<td>8</td>
<td>3</td>
<td>1</td>
<td>97</td>
<td>97</td>
<td>0</td>
</tr>
<tr>
<td>9</td>
<td>3</td>
<td>2</td>
<td>60</td>
<td>97</td>
<td>-37</td>
</tr>
<tr>
<td>10</td>
<td>3</td>
<td>3</td>
<td>96</td>
<td>97</td>
<td>-1</td>
</tr>
</tbody>
</table>

In this example the programmer used visit 1 as baseline. However, for some reason one patient had two visit 1 observations. So when baseline values were selected this patient ended up with two observations instead of one, and when change from baseline was calculated the results were not as expected. It can be seen that for all other patients the single baseline value is repeated in every observation and change is calculated from that single value. However, the patient with the repeated visit one value only has the second value repeated on all further observations, resulting in change being calculated from two different baseline values. It should also be noted that the order of the visit 1 values are there by chance, and so if the data was sorted differently then the value 77 would have been carried forward instead of 73.

If "repeats of BY values" appears in the LOG, one should check the datasets to see which dataset has duplicate records per BY variable, and then decide what should be done so there is only one observation per by variable. In this example, perhaps there was a data issue and one of them is a repeated measure, and so a rule should be specified that perhaps the second value based on date and time should be used as the baseline. If there is no date and time then the duplicates can still be identified so that one value can be selected manually as the baseline value. However, in most cases it should be possible to add another by variable to identify which observation was recorded first, and a programmed solution as shown below is always best.

```
DATA dups nodups;
  SET baseline;
  BY ptno visit;
  IF NOT(FIRST.ptno AND LAST.ptno) THEN OUTPUT dups;
  IF LAST.ptno THEN OUTPUT nodups;
RUN;

TITLE3 'Duplicate records - only the last observation will be used';
PROC PRINT DATA=dups;
RUN;
```
VARIABLE X WILL BE OVERWRITTEN
This is where two or more datasets are merged and they have variables in common which are not in the BY statement. It results in the values of the second dataset overwriting the values of the first dataset from common variables which are not in the BY statement. This is an error which hopefully does not go unnoticed, but it is difficult to know how often it occurs. The worrying thing here is that it is a message which perhaps is not seen as often as it should be. This is not because it does not occur often, but because the default option used by most users will not print it in the LOG. OPTIONS MSGLEVEL=N is used by most users so that all notes are printed in the LOG. However, the message in the LOG informing the user when variables are being overwritten is only printed when OPTION MSGLEVEL=I is used.

A classic example where this can happen is selecting baselines based on a visit and then merging this new dataset with the original dataset to calculate change. If the programmer forgets to remove the visit variable from the baseline dataset, the baseline visit value will replace the visit value in the original dataset after the merge if the baseline dataset is the second dataset to be specified in the MERGE statement. However, as no messages are sent to the LOG, the user will be unaware of this unless they spot it when checking the output.

Program:

```plaintext
OPTIONS MSGLEVEL=N;

DATA wt_by_vis;
  DO ptno=1 TO 3;
    DO visit=1 TO 3;
      weight=60+MOD(INT((RANUNI(0)*1000)),40);
      OUTPUT;
    END;
  END;
RUN;

TITLE3 'Weight by visit';
PROC PRINT DATA=wt_by_vis;
RUN;
TITLE3;

DATA baseline;
  SET wt_by_vis;
  WHERE visit=2;
RUN;

DATA change;
  MERGE wt_by_vis baseline(RENAME=(weight=baseline));
  BY ptno;
  change=weight-baseline;
RUN;

TITLE3 'CHANGE from baseline - VISIT is overwritten for the first observation of each patient';
TITLE4 'but nothing is mentioned in the LOG';
PROC PRINT DATA=change;
RUN;
TITLE3;
TITLE4;
```
Log:

NOTE: The data set WORK.WT_BY_VIS has 9 observations and 3 variables.

NOTE: There were 9 observations read from the data set WORK.WT_BY_VIS.

NOTE: There were 3 observations read from the data set WORK.WT_BY_VIS.

NOTE: The data set WORK.BASELINE has 3 observations and 3 variables.

DATA change;
  MERGE wt_by_vis baseline(RENAME=(weight=baseline));
  BY ptno;
    change=weight-baseline;
RUN;

NOTE: There were 9 observations read from the data set WORK.WT_BY_VIS.

NOTE: There were 3 observations read from the data set WORK.BASELINE.

NOTE: The data set WORK.CHANGE has 9 observations and 5 variables.

NOTE: There were 9 observations read from the data set WORK.CHANGE.

Output:

Weight by visit

<table>
<thead>
<tr>
<th>Obs</th>
<th>PTNO</th>
<th>VISIT</th>
<th>WEIGHT</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1</td>
<td>1</td>
<td>79</td>
</tr>
<tr>
<td>2</td>
<td>1</td>
<td>2</td>
<td>76</td>
</tr>
<tr>
<td>3</td>
<td>1</td>
<td>3</td>
<td>65</td>
</tr>
<tr>
<td>4</td>
<td>2</td>
<td>1</td>
<td>66</td>
</tr>
<tr>
<td>5</td>
<td>2</td>
<td>2</td>
<td>60</td>
</tr>
<tr>
<td>6</td>
<td>2</td>
<td>3</td>
<td>90</td>
</tr>
<tr>
<td>7</td>
<td>3</td>
<td>1</td>
<td>79</td>
</tr>
<tr>
<td>8</td>
<td>3</td>
<td>2</td>
<td>92</td>
</tr>
<tr>
<td>9</td>
<td>3</td>
<td>3</td>
<td>75</td>
</tr>
</tbody>
</table>

CHANGE from baseline - VISIT is overwritten for the first observation of each patient
but nothing is mentioned in the LOG

<table>
<thead>
<tr>
<th>Obs</th>
<th>PTNO</th>
<th>VISIT</th>
<th>WEIGHT</th>
<th>BASELINE</th>
<th>CHANGE</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1</td>
<td>2</td>
<td>79</td>
<td>76</td>
<td>3</td>
</tr>
<tr>
<td>2</td>
<td>1</td>
<td>2</td>
<td>76</td>
<td>76</td>
<td>0</td>
</tr>
<tr>
<td>3</td>
<td>1</td>
<td>3</td>
<td>65</td>
<td>76</td>
<td>-11</td>
</tr>
<tr>
<td>4</td>
<td>2</td>
<td>2</td>
<td>66</td>
<td>60</td>
<td>6</td>
</tr>
<tr>
<td>5</td>
<td>2</td>
<td>2</td>
<td>60</td>
<td>60</td>
<td>0</td>
</tr>
<tr>
<td>6</td>
<td>2</td>
<td>3</td>
<td>90</td>
<td>60</td>
<td>30</td>
</tr>
<tr>
<td>7</td>
<td>3</td>
<td>2</td>
<td>79</td>
<td>92</td>
<td>-13</td>
</tr>
<tr>
<td>8</td>
<td>3</td>
<td>2</td>
<td>92</td>
<td>92</td>
<td>0</td>
</tr>
<tr>
<td>9</td>
<td>3</td>
<td>3</td>
<td>75</td>
<td>92</td>
<td>-17</td>
</tr>
</tbody>
</table>

It can be seen here that Visit=1 of all patients is being overwritten by the baseline visit value from the baseline dataset. So if a summary is performed later on, the results for visit 2 will also include the values of visit 1 without the user being aware of it.

If OPTIONS MSGLEVEL=I is used then although the result will be the same, the following message will appear in the LOG.
OPTIONS MSGLEVEL=I;

DATA change;
    MERGE wt_by_vis baseline(RENAME=(weight=baseline));
    BY ptno;
    change=weight-baseline;
RUN;

INFO: The variable VISIT on data set WORK.WT_BY_VIS will be overwritten by data set WORK.BASELINE.
NOTE: There were 9 observations read from the data set WORK.WT_BY_VIS.
NOTE: There were 3 observations read from the data set WORK.BASELINE.
NOTE: The data set WORK.CHANGE has 9 observations and 5 variables.

The solution to this problem is very simple and is probably what most good programming practice guides recommend, only keep variables which are required. In this example the VISIT variable is not required once the baseline has been selected, so by dropping this variable the problem will be solved.

DATA change;
    MERGE wt_by_vis baseline(KEEP=ptno weight RENAME=(weight=baseline));
    BY ptno;
    change=weight-baseline;
RUN;

NOTE: There were 9 observations read from the data set WORK.WT_BY_VIS.
NOTE: There were 3 observations read from the data set WORK.BASELINE.
NOTE: The data set WORK.CHANGE has 9 observations and 5 variables.

TITLE3 'CHANGE from baseline - only keep required variables';
PROC PRINT DATA=change;
RUN;

NOTE: There were 9 observations read from the data set WORK.CHANGE.

CHANGE from baseline - only keep required variables

<table>
<thead>
<tr>
<th>Obs</th>
<th>PTNO</th>
<th>VISIT</th>
<th>WEIGHT</th>
<th>BASELINE</th>
<th>CHANGE</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1</td>
<td>1</td>
<td>79</td>
<td>76</td>
<td>3</td>
</tr>
<tr>
<td>2</td>
<td>1</td>
<td>2</td>
<td>76</td>
<td>76</td>
<td>0</td>
</tr>
<tr>
<td>3</td>
<td>1</td>
<td>3</td>
<td>65</td>
<td>76</td>
<td>-11</td>
</tr>
<tr>
<td>4</td>
<td>2</td>
<td>1</td>
<td>66</td>
<td>60</td>
<td>6</td>
</tr>
<tr>
<td>5</td>
<td>2</td>
<td>2</td>
<td>60</td>
<td>60</td>
<td>0</td>
</tr>
<tr>
<td>6</td>
<td>2</td>
<td>3</td>
<td>90</td>
<td>60</td>
<td>30</td>
</tr>
<tr>
<td>7</td>
<td>3</td>
<td>1</td>
<td>79</td>
<td>92</td>
<td>-13</td>
</tr>
<tr>
<td>8</td>
<td>3</td>
<td>2</td>
<td>92</td>
<td>92</td>
<td>0</td>
</tr>
<tr>
<td>9</td>
<td>3</td>
<td>3</td>
<td>75</td>
<td>92</td>
<td>-17</td>
</tr>
</tbody>
</table>

The VISIT variable now has the correct values in all observations, and the LOG does not show any messages which suggests there may be issues.
W.D FORMAT WAS TOO SMALL
This is another comment in the LOG which is not too uncommon. It basically means that a format used to print a number is too small, and so SAS ignores the format for that value and fits it as best as possible. At best this just means the loss of a decimal place, but at worse it can lead to an Exx number instead of an actual number appearing in a summary table or a listing. Although these may be picked up in a small summary tables, in large tables and listings where not all values are checked, the errors may go unnoticed.

Program:

DATA wd_fmt;
  a=123456.45;
  x=a;
  y=a;
  z=a;
  FORMAT x 5.2 y 6.2 z 8.2;
  LABEL a = "A - no format"
  x = "X - format 5.2"
  y = "Y - format 6.2"
  z = "Z - format 8.2";
RUN;

TITLE3 "ALL variables have the same value, but the formats affect how they look";
PROC PRINT DATA=wd_fmt LABEL;
RUN;
TITLE3;

Log:

DATA wd_fmt;
  a=123456.45;
  x=a;
  y=a;
  z=a;
  FORMAT x 5.2 y 6.2 z 8.2;
  LABEL a = "A - no format"
  x = "X - format 5.2"
  y = "Y - format 6.2"
  z = "Z - format 8.2";
RUN;

NOTE: The data set WORK.WD_FMT has 1 observations and 4 variables.

TITLE3 "ALL variables have the same value, but the formats affect how they look";
PROC PRINT DATA=wd_fmt LABEL;
RUN;

NOTE: There were 1 observations read from the data set WORK.WD_FMT.
NOTE: At least one W.D format was too small for the number to be printed. The decimal may be shifted by the 'BEST' format.
This illustration shows how the original value as presented in A can be presented in tables or listings if the correct format is not used. The only way to know that the correct format is not used is the message in the LOG, and so it is very important that this message is not ignored. The only problem with SAS at the moment is that it does not mention which variable is the cause, and so if many variables are being listed or appear in a dataset, then they all have to be checked to determine which variable has the format issue. This can be done using PROC FREQ of all numeric variables, so that all values are listed and a quick scan should highlight which have problems.

DATA wd_fmt2;
  DO i=1 TO 10;
    a=RANUNI(0)*2000;
    b=RANUNI(0)*100;
    OUTPUT;
  END;
  FORMAT a 3.1 b 4.1;
RUN;

NOTE: The data set WORK.WD_FMT2 has 10 observations and 3 variables.

TITLE3 'Data with an format which is too small';
PROC PRINT DATA=wd_fmt2;
RUN;

NOTE: There were 10 observations read from the data set WORK.WD_FMT2.
NOTE: At least one W.D format was too small for the number to be printed. The decimal may be shifted by the 'BEST' format.

TITLE3 'Check which value is incorrect';
PROC FREQ DATA=wd_fmt2;
  TABLES a b;
RUN;

NOTE: There were 10 observations read from the data set WORK.WD_FMT2.
NOTE: At least one W.D format was too small for the number to be printed. The decimal may be shifted by the 'BEST' format.

Data with an format which is too small

<table>
<thead>
<tr>
<th>Obs</th>
<th>I</th>
<th>A</th>
<th>B</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1</td>
<td>2E3</td>
<td>94.1</td>
</tr>
<tr>
<td>2</td>
<td>2</td>
<td>2E3</td>
<td>9.9</td>
</tr>
<tr>
<td>3</td>
<td>3</td>
<td>1E3</td>
<td>53.1</td>
</tr>
<tr>
<td>4</td>
<td>4</td>
<td>505</td>
<td>48.0</td>
</tr>
<tr>
<td>5</td>
<td>5</td>
<td>2E3</td>
<td>41.6</td>
</tr>
<tr>
<td>6</td>
<td>6</td>
<td>2E3</td>
<td>68.7</td>
</tr>
<tr>
<td>7</td>
<td>7</td>
<td>2E3</td>
<td>76.2</td>
</tr>
<tr>
<td>8</td>
<td>8</td>
<td>998</td>
<td>17.2</td>
</tr>
<tr>
<td>9</td>
<td>9</td>
<td>2E3</td>
<td>4.4</td>
</tr>
<tr>
<td>10</td>
<td>10</td>
<td>1E3</td>
<td>47.1</td>
</tr>
</tbody>
</table>
Check which value is incorrect

The FREQ Procedure

<table>
<thead>
<tr>
<th></th>
<th>Frequency</th>
<th>Percent</th>
<th>Cumulative Frequency</th>
<th>Percent</th>
</tr>
</thead>
<tbody>
<tr>
<td>505</td>
<td>1</td>
<td>10.00</td>
<td>1</td>
<td>10.00</td>
</tr>
<tr>
<td>998</td>
<td>1</td>
<td>10.00</td>
<td>2</td>
<td>20.00</td>
</tr>
<tr>
<td>1E3</td>
<td>2</td>
<td>20.00</td>
<td>4</td>
<td>40.00</td>
</tr>
<tr>
<td>2E3</td>
<td>6</td>
<td>60.00</td>
<td>10</td>
<td>100.00</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th></th>
<th>Frequency</th>
<th>Percent</th>
<th>Cumulative Frequency</th>
<th>Percent</th>
</tr>
</thead>
<tbody>
<tr>
<td>4.4</td>
<td>1</td>
<td>10.00</td>
<td>1</td>
<td>10.00</td>
</tr>
<tr>
<td>9.9</td>
<td>1</td>
<td>10.00</td>
<td>2</td>
<td>20.00</td>
</tr>
<tr>
<td>17.2</td>
<td>1</td>
<td>10.00</td>
<td>3</td>
<td>30.00</td>
</tr>
<tr>
<td>41.6</td>
<td>1</td>
<td>10.00</td>
<td>4</td>
<td>40.00</td>
</tr>
<tr>
<td>47.1</td>
<td>1</td>
<td>10.00</td>
<td>5</td>
<td>50.00</td>
</tr>
<tr>
<td>48.0</td>
<td>1</td>
<td>10.00</td>
<td>6</td>
<td>60.00</td>
</tr>
<tr>
<td>53.1</td>
<td>1</td>
<td>10.00</td>
<td>7</td>
<td>70.00</td>
</tr>
<tr>
<td>68.7</td>
<td>1</td>
<td>10.00</td>
<td>8</td>
<td>80.00</td>
</tr>
<tr>
<td>76.2</td>
<td>1</td>
<td>10.00</td>
<td>9</td>
<td>90.00</td>
</tr>
<tr>
<td>94.1</td>
<td>1</td>
<td>10.00</td>
<td>10</td>
<td>100.00</td>
</tr>
</tbody>
</table>

Here we can see that variable A is the problem, and so changing the format of A will remove the note from the LOG.

```
DATA wd_fmt2;
  DO i=1 TO 10;
    a=RANUNI(0)*2000;
    b=RANUNI(0)*100;
    OUTPUT;
  END;
  FORMAT a 6.1 b 4.1;
RUN;
```

NOTE: The data set WORK.WD_FMT2 has 10 observations and 3 variables.

```
TITLE3 'Data with the correct format';
PROC PRINT DATA=wd_fmt2;
RUN;
```

NOTE: There were 10 observations read from the data set WORK.WD_FMT2.

Data with the correct format

<table>
<thead>
<tr>
<th>Obs</th>
<th>I</th>
<th>A</th>
<th>B</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1</td>
<td>918.6</td>
<td>7.2</td>
</tr>
<tr>
<td>2</td>
<td>2</td>
<td>485.1</td>
<td>72.5</td>
</tr>
<tr>
<td>3</td>
<td>3</td>
<td>561.0</td>
<td>62.9</td>
</tr>
<tr>
<td>4</td>
<td>4</td>
<td>1593.8</td>
<td>17.4</td>
</tr>
<tr>
<td>5</td>
<td>5</td>
<td>1966.5</td>
<td>1.0</td>
</tr>
<tr>
<td>6</td>
<td>6</td>
<td>1868.3</td>
<td>32.6</td>
</tr>
<tr>
<td>7</td>
<td>7</td>
<td>878.8</td>
<td>81.5</td>
</tr>
<tr>
<td>8</td>
<td>8</td>
<td>1708.4</td>
<td>63.3</td>
</tr>
<tr>
<td>9</td>
<td>9</td>
<td>669.3</td>
<td>54.2</td>
</tr>
<tr>
<td>10</td>
<td>10</td>
<td>1689.1</td>
<td>26.7</td>
</tr>
</tbody>
</table>
CONCLUSION
As demonstrated in this paper, it is very important that programmers use all options open to them to ensure problems are highlighted, and any messages other than the standard NOTE stating how many observations are read in or sent to the output dataset is investigated. Failure to investigate and resolve these issues can have serious consequences on the results those programs produce.

It is also important to note that these issues will not be obvious if the programs are validated simply by checking the code, and so outputs must be checked very carefully in order to spot these issues. Using OPTIONS MSGLEVEL=I is also highly recommended to be used as default because it only sends messages to the LOG when there is a problem, which once resolved will not have any other impact.

CONTACT INFORMATION
Your comments and questions are valued and encouraged. Contact the author at:

Author Name : Shafi Chowdhury
Company : Shafi Consultancy Limited
Address : 7 Blossom Way, Uxbridge, Middlesex, UB10 9LL, U.K.
Phone : +44 770 288 7219
Email : Shafi@shaficonsultancy.com
Web : www.shaficonsultancy.com

Brand and product names are trademarks of their respective companies.